

# SPARQL

Petr Křemen

October 19, 2017

## 1 Exercises – DBPedia

Use the <http://dbpedia.org/sparql> endpoint for posing queries. For finding suitable resources, please use either the DBPedia ontology <http://mappings.dbpedia.org/server/ontology/classes/>, or the browser available at [http://live.dbpedia.org/page/Czech\\_Republic](http://live.dbpedia.org/page/Czech_Republic).

**Ex. 1** — Find all types (classes), the resource describing Prague belongs to.

**Answer (Ex. 1)** —

```
SELECT DISTINCT ?c
WHERE
{
  <http://dbpedia.org/resource/Prague> a ?c
}
```

**Ex. 2** — Find labels of Prague in english and german.

**Answer (Ex. 2)** —

```
SELECT DISTINCT ?l
WHERE
{
  <http://dbpedia.org/resource/Prague> rdfs:label ?l.
  FILTER (lang(?l) in ("en", "de"))
}
```

**Ex. 3** — List all EU members (countries) and show what is the relative strength of a voter in different EU countries (i.e. the ratio of total population to the number of seats in the EU parliament). Take the Czech Republic ([http://dbpedia.org/resource/Czech\\_Republic](http://dbpedia.org/resource/Czech_Republic)) as an example to explore the properties. Note that not all information is available for all countries.

Answer (Ex. 3) —

```
PREFIX dct: <http://purl.org/dc/terms/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbrc: <http://dbpedia.org/resource/Category:>
SELECT DISTINCT ?country ?seats xsd:integer(?populationTotal) ?ratio
{
  ?country dct:subject dbrc:Member_states_of_the_European_Union ;
    a dbo:Country .

  OPTIONAL {
    ?country dbp:euseats ?seats .
  }

  OPTIONAL {
    ?country dbo:populationTotal ?populationTotal .
  }

  BIND ((xsd:double(?seats)/xsd:double(?populationTotal)) AS ?ratio)
}
```

Ex. 4 — How many actors died in another country than they got born according to DBPedia. Group by countries where the actors were born.

Answer (Ex. 4) —

```
PREFIX dct: <http://purl.org/dc/terms/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbp: <http://dbpedia.org/property/>
SELECT ?bp (count(*) AS ?count)
{
  ?p dbo:occupation <http://dbpedia.org/resource/Actor> ;
    dbo:deathPlace ?dp ;
    dbo:birthPlace ?bp .
  ?dp a dbo:Country .
  ?bp a dbo:Country .
  FILTER (?dp != ?bp)
} GROUP BY ?bp
ORDER BY desc(?count)
```

Ex. 5 — Which world countries have more than one official language ?

Answer (Ex. 5) —

```

PREFIX dbo: <http://dbpedia.org/ontology/>
SELECT ?p (count(?l) AS ?languageCount)
{
  ?p dbo:officialLanguage ?l
}
GROUP BY ?p
HAVING (count(?l) > 1)
ORDER BY desc(count(?l))

```

**Ex. 6** — Find ten countries with the highest population and show their gross domestic product per capita.

**Answer (Ex. 6)** —

```

PREFIX dct: <http://purl.org/dc/terms/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbp: <http://dbpedia.org/property/>
SELECT DISTINCT ?country ?cnt (xsd:integer(?gdp) AS ?gdpPerCapita)
WHERE
{
  ?country a dbo:Country .
  ?country dbp:populationCensus ?cnt .
  ?country dbp:gdpNominalPerCapita ?gdp .
}
ORDER BY DESC(?cnt)
LIMIT 10

```

**Ex. 7** — Find all parliamentary republics optionally together with their english label (if exist)

**Answer (Ex. 7)** —

```

PREFIX dct: <http://purl.org/dc/terms/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbr: <http://dbpedia.org/resource/>
SELECT DISTINCT ?country ?l
WHERE
{
  ?country a dbo:Country .
  ?country dbo:governmentType dbr:Parliamentary_republic .
  OPTIONAL {
    ?country rdfs:label ?l
  }
}

```

```
    FILTER (lang(?l)="en")
  }
}
```

**Ex. 8** — Find all female ancestors of the Prince Charles.

**Answer (Ex. 8)** —

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbr: <http://dbpedia.org/resource/>

SELECT DISTINCT ?s WHERE {
  <http://dbpedia.org/resource/Charles,_Prince_of_Wales> dbo:parent+ ?s .
  ?s <http://xmlns.com/foaf/0.1/gender> "female"@en .
}
```

## 2 Relevant References

- DBPedia SPARQL endpoint <http://dbpedia.org/sparql>
- OSW ontology SPARQL endpoint <http://onto.fel.cvut.cz:7300/repositories/osw2017-ontology>
- SPARQL endpoint list <http://www.w3.org/wiki/SparqlEndpoints>